


# VB.NET

コンポーネント活用講座  
～ 教養課程 ～

## スモールパーツ プログラミング



第28回

小森 大輔 KOMORI, Daisuke

### Print関連のコンポーネント —Part2—

Small Parts Programming

#### Technology Tools

- Visual Basic .NET
- Visual C# .NET
- SQL Server 2000
- Oracle 9i
- Access 2002
- ASP.NET
- Internet Information Services
- Other:

#### Level



#### Samples

・この記事で取り上げたソースコードおよびサンプルプログラムは、付録CD-ROMの¥DOTNET¥SMALLディレクトリに収録しています。

¥SAMPLE01  
PrintPreviewControlを使った印刷プレビュー


¥SAMPLE02  
複数ページのプレビュー

¥SAMPLE03  
PrintPreviewDialogを使った印刷プレビュー

¥SAMPLE04  
PrintDialogを使った印刷ダイアログボックスの利用

¥SAMPLE05  
PageSetupDialogを使ったページ設定ダイアログボックスの利用


#### 印刷設定とプレビュー



本連載は、Visual Basic .NET (以下VB.NET) の標準コントロールにスポットを当て、とことん使い込んで、その機能や役割を探ってゆきます。

今回はPrintDocumentコントロールを使用した「プリンタへの出力処理の基本」を詳しく説明しました。今回は他のコントロールも使用して、ページ設定やプリントプレビューについて見ていきたいと思えます。

#### PrintPreview Control コントロール



印刷機能を持つアプリケーションでユーザーが印刷を行なう際、どのように印刷されるのかプレビュー画面で確認したいというのはよくある場面です。

PrintPreviewControlは、印刷プレビューを行なうことができるコントロールです。PrintPreviewControlコントロールをフォームに配置し、このコントロールのDocumentプロパティにPrintDocumentコントロールのインスタンスを指定するだけで印刷プレビューを行なうことができます。

リスト1は、テキストファイルを読み込んで印刷出力するサンプルの、プレビューを行なうコードです。

これは、前号で印刷出力を行なったサンプル (PrintSample01) を変更したプログラムで、ButtonのクリックイベントでPrintPreviewControlのDocumentプロパティに、PrintDocumentのインスタンスを指定しています。あとは通常の印刷出力と同じように、PrintDocumentのBeginPrintイベント、PrintPageイベント、EndPrintイベントが発生するので、この各イベントプロシージャの中でプリンタへ出力する場合と同じ処理を行なえば、

リスト1：印刷プレビュー

```

' ファイルを読み込むStreamReaderを宣言
Private sr As StreamReader

' サンプルテキストファイルのパス
Private strFilePath As String

Private Sub Button1_Click(ByVal sender As System.Object, _
    ByVal e As System.EventArgs) Handles Button1.Click
    ' プレビューを開始
    PrintPreviewControl1.Document = PrintDocument1
End Sub

Private Sub PrintDocument1_BeginPrint(ByVal sender As Object, _
    ByVal e As System.Drawing.Printing.PrintEventArgs) _
    Handles PrintDocument1.BeginPrint
    Try
        ' ファイルと shift_jis のエンコーディングを指定して
        ' StreamReaderのインスタンスを作成
        sr = New StreamReader(strFilePath & "%Sample.txt", _
            Encoding.GetEncoding("shift_jis"))

    Catch ex As Exception
        MessageBox.Show(ex.Message)
    End Try
End Sub

Private Sub PrintDocument1_PrintPage(_
    ByVal sender As Object, _
    ByVal e As System.Drawing.Printing.PrintPageEventArgs) _
    Handles PrintDocument1.PrintPage
    ' StreamReaderのインスタンスが作成されていない場合は抜ける
    If IsNothing(sr) Then Exit Sub

    ' 文字列を描画するブラシを作成
    Dim b As Brush = New SolidBrush(Color.Black)

    ' PageSettingsのマージンから出力位置の初期値を格納
    Dim posX As Single = e.PageSettings.Margins.Left
    Dim posY As Single = e.PageSettings.Margins.Top

    ' ストリームから文字列を1行読み取り
    Dim strDat As String = sr.ReadLine

    ' 最終行まで読み込み
    Do Until strDat Is Nothing

        ' Fontの高さからY方向出力位置を算出
        posY = posY + Font.GetHeight(e.Graphics)

        ' PrintPageEventArgsのGraphicsへ文字列を描画
        e.Graphics.DrawString(strDat, Font, b, posX, posY)

        strDat = sr.ReadLine

    Loop
    sr.Close()
End Sub

Private Sub PrintDocument1_EndPrint(ByVal sender As Object, _
    ByVal e As System.Drawing.Printing.PrintEventArgs) _
    Handles PrintDocument1.EndPrint
    ' ストリームリソースの解放
    If Not IsNothing(sr) Then sr.Close()
End Sub

```



PrintPreviewControlのビュー内へ出力した内容が表示されます。

図1は、このサンプルを実行した画面です。

## PrintPreviewControlのプロパティ

PrintPreviewControlコントロールに、独自に実装されたプロパティはそれほど多くありません。これらの利用法、注意点などについて見ていきましょう。

### Zoomプロパティ

このプロパティはページの表示倍率を表わし、“1.0”が表示100%の値となります。

リスト2はプレビュー部分のクリックで表示倍率を拡大縮小するコード例で、図2はこのコードの実行画面です。

このコードを試しているときに気が付いたのですが、Zoomプロパティを変更すると、プレビューしたページがPrintPreviewControlのサイズ (Width, Height) に収まる状態と収まらない状態に切り替わる際に、実際にコ

ントロールのサイズは変更されていないにもかかわらず、なぜかResizeイベントが発生します。

ResizeイベントでZoomを切り替えるような処理を行なう際には、この点に注意してください。

図1：PrintPreviewControlを使用した印刷プレビュー

